
Pythagorean points (X,Y): X^2+Y^2=Z^2 (X,Y,Z integers) in red
Other points (X,Y) more or less green depending on how far Z is from an integer.
See curved traces between neighbours.

Recently I extended graphical capabilities of the Survo system [4] by a feature
that makes possible to define the color of a point in a scatter diagram of variables,
say X and Y , by integer values of any variable, say F , in the current data set.

As one illustration of this enhancement I decided to make such a plot of variables
X, Y having integer values 1, 2, . . . , 200 and color a point (X, Y ) into ’red’ if X and
Y belong to a Pythagorean triple of integers X, Y, Z where Z2 = X2 + Y 2. Other
points would be colored on a scale from ’red’ to ’green’ depending on how much
Z =

√
X2 + Y 2 differs from the nearest integer on a nonlinear scale on the interval

(0, 0.5) (see details in the appendix).
The picture on the first page tells more than displays on the same topic of the

following style shown e.g. in [2].

Since in the colored graph ‘almost Pythagorean’ points become more or less ‘red-
dish’, interesting traces connecting true Pythagorean points will be seen.

This illustration revealed to me e.g. that all Pythagorean points (X, Y ) seem to
be located on hyperbolas

Y = 2C(X − C)/(X − 2C), C = 1, 2, . . . .

To get them plotted, their equations can be written in parametric form

(1) X(t) = C
√

2/ tan(t) + 2C, Y (t) = C
√

2 tan(t) + 2C, 0 < t < π/2

and now these curves will seen (in yellow) for values C=1,2,...,50 in the next graph.

Obviously the number N(C) of Pythagorean points on the curve

Y = 2C(X − C)/(X − 2C)

is finite and by direct computation we get for C = 1, 2, . . . , 16
C 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16
N(C) 2 4 6 6 6 12 6 8 10 12 6 18 6 12 18 10

Numbers N(C)/2 seem to form the sequence A078644 in Sloane’s Encyclopedia
of Integer Sequences [3] and thus the general expression for N(C) is

N(C) = σ0(2C2)

where σ0(n) is the number of divisors of n.

Pythagorean points (X,Y): X^2+Y^2=Z^2 (X,Y,Z integers) in red
Other points (X,Y) more or less green depending on how far Z is from an integer.
Yellow hyperbolas connecting Pythagorean points

We prove this statement in the following way:

Since X(t) and Y (t) must be positive integers, expressions
√

2/ tan(t) and
√

2 tan(t)
must be rational and this implies that tan(t) = 2k2 where k is a rational number.

Then equations (1) can be written in the form

(2) X(k) = 2C + C/k,

(3) Y (k) = 2C + 2Ck.

The last member of a Pythagorean triple Z(k) (Z(k)2 = X(k)2 + Y (k)2) is

(4) Z(k) = 2Ck + 2C + C/k = X(k) + Y (k) − 2C.

Then N(C) will be the number of k-values for which all three expressions (2), (3),
(4) are positive integers.

Since, according to (4), Z(k) is an integer when X(k) and Y (k) are integers, it
is sufficient to pay attention to X(k) and Y (k) only.

If C is a prime number p 6= 2, it is seen immediately that the only accepted
values of k, giving integer values for X(k), Y (k) i.e. simultaneously for C/k and
2Ck, are

k = 1, p, 1/p

and their halves
k = 1/2, p/2, 1/2p

since the latter ones give an integer value also to 2Ck.
Similarly, if C is product of two primes p, q 6= 2, the only valid values for k are

k = 1, p, q, pq, p/q, q/p, 1/p, 1/q, 1/pq

and their halves

k = 1/2, p/2, q/2, pq/2, p/2q, q/2p, 1/2p, 1/2q, 1/2pq.

Now it is crucial to notice that there is a bijection from divisors of 2C2 = 2p2q2 to
these values of k by

2ipipqiq ⇔ 2i−1pip−1qiq−1

where i = 0, 1, ip, iq = 0, 1, 2.
In the general case let the prime factorization of C be C = 2apa1

1
. . . pam

m . Then
the bijection is

(5) 2ipi1
1

. . . pim

m ⇔ 2i−a−1pi1−a1

1
. . . pim−am

m

where i = 0, 1, . . . , 2a + 1, i1 = 0, 1, . . . , 2a1, . . . , im = 0, 1, ..., 2am.
Since the only valid k-values (making X(k) and Y (k) integers) are values on the
right-hand side of (5) and the left-hand side gives all divisors of 2C2, we finally
obtain

N(C) = σ0(2C2).

The same result has been recently derived by McCullough [1] but our treatment
of the problem is simpler due to the parametric representation (2),(3) while Mc-
Cullough is operating in a more general framework. Also the nice interpretation of
N(C) as the number of divisors of 2C2 seems not to be mentioned in his paper.

From (4) we get the excess X(T ) + Y (T ) − Z(T ) = 2C and this is the basis of
enumeration in McCullough’s paper. Since the excess is always an even integer, it
is clear that only integer values of C give Pythagorean triples. Hence equations
(2), (3), (4) give an exhaustive enumeration of these triples for C = 1, 2, . . . and
for various k-values obtained from divisors of 2C2 according to (5).

Geometric interpretation (17 June 2005)

As a personal communication Darryl McCullough wrote me (11 June 2005):
“C is the radius of the incircle, so place the incircle in the first quadrant tangent to
the x- and y-axes at x = C and y = C. Then, the hyperbola is the locus of points
(X, Y ) such that X and Y are the intercepts of a line tangent to the incircle.”

Thus N(C) is also the number of Pythagorean triples with a given radius C of the
incircle.

It can be concluded further that the hyperbola has its loci in points (0, 0) and
(4C, 4C) and the difference of distances of any point on the hyperbola from these
loci is 4C. Then the circles X2 + Y 2 = (4C)2 and (X − X0)

2 + (Y − Y0)
2 = R2,

where (X0, Y0) are the legs of a right triangle with the radius C of the incircle and
R2 = (X0 − 4C)2 + (Y0 − 4C)2, are mutually tangent and the line from (0, 0) to
(X0, Y0) goes through the tangent point of those circles.

The following graph illustrates the case C = 3 giving six Pythagorean triples
(7, 24, 15), (8, 15, 17), (9, 12, 25), (12, 9, 15), (15, 8, 17), (24, 7, 25).

Appendix:

Making the graphics and calculations as Survo applications

As I mentioned earlier, this study is a consequence of a small enhancement made
in the Survo system [4]. It is worthwhile to show how Survo has given support
for making the graphics and calculations needed in this presentation. In fact, the
entire manuscript has been prepared with the Survo editor into LATEXformat.

The grid of points {(X, Y ) | X, Y = 1, 2, . . . , 200} is created as a Survo data file
TRIPLES and variables Z, D (deviation of Z from the nearest integer), and F (the
color index) are computed in the Survo edit field as follows:

____________________________________________________________________

Creating a data file TRIPLES:

FILE CREATE TRIPLES,64,8

FIELDS:

1 N 2 X

2 N 2 Y

3 N 4 Z

END

FILE INIT TRIPLES,40000 / Space for 200*200=40000 observations

Creating a 200*200 grid of points (X,Y):

M=200

X=int((ORDER-1)/M)+1

Y=ORDER-X*M+M

VAR X,Y TO TRIPLES / Computing X and Y values

Computing Z values:

VAR Z=sqrt(X*X+Y*Y) TO TRIPLES

Computing deviation D from the nearest integer:

D=min(Z-int(Z),int(Z+1)-Z)

VAR D TO TRIPLES

Non-linear mapping of D values to color indices -21,-20,...,-1:

F=-int(21*(2*D)^0.2)-1

VAR F:2 TO TRIPLES

____________________________________________________________________

The picture on the first page is plotted as a PostScript file PYTH1.PS by the
following Survo PLOT scheme:

____________________________________________________________________

PLOT TRIPLES,X,Y / DEVICE=PS,PYTH1.PS

POINT=6,4 POINT_COLOR=F (This is the "main point"!)

SCALE=0.5:?,1,200,200.5:? XLABEL= YLABEL= HEADER=

SIZE=1270,1270

PEN=[Swiss(7.26)]

XDIV=100,1200,100 YDIV=100,1200,100

TEXTS=T1,T2,T3

T1=[RED],Pythagorean_points_(X;Y):_X^2+Y^2=Z^2_&

(X;Y;Z_integers)_in_red,91,1234

T2=[BLACK],Other_points_(X;Y)_more_or_less_green_&

depending_on_how_far_Z_is_from_an_integer.,91,1211

T3=[BLACK],See_curved_traces_between_neighbours.,91,1188

21 colors gradually from red to green:(cmyk codes)

FILL(-1)=0.00,1.00,1.0,0 FILL(-11)=0.40,0.54,0.8,0

FILL(-2)=0.13,0.90,0.8,0 FILL(-12)=0.43,0.50,0.8,0

FILL(-3)=0.16,0.86,0.8,0 FILL(-13)=0.46,0.46,0.8,0

FILL(-4)=0.19,0.82,0.8,0 FILL(-14)=0.49,0.42,0.8,0

FILL(-5)=0.22,0.78,0.8,0 FILL(-15)=0.52,0.38,0.8,0

FILL(-6)=0.25,0.74,0.8,0 FILL(-16)=0.55,0.34,0.8,0

FILL(-7)=0.28,0.70,0.8,0 FILL(-17)=0.58,0.30,0.8,0

FILL(-8)=0.31,0.66,0.8,0 FILL(-18)=0.61,0.26,0.8,0

FILL(-9)=0.34,0.62,0.8,0 FILL(-19)=0.64,0.22,0.8,0

FILL(-10)=0.37,0.58,0.8,0 FILL(-20)=0.67,0.18,0.8,0

FILL(-21)=0.70,0.14,0.8,0

____________________________________________________________________

The hyperbolas (1) are plotted in yellow for C = 1, 2, . . . , 50 as a PostScript
file PYTH2.PS and combined with PYTH1B.PS, a modified version (partially different
texts) of the first figure:

____________________________________________________________________

Plotting the hyperbolas:

PLOT X(t)=C*sqrt(2/tan(t))+2*C-1/4,Y(t)=C*sqrt(2*tan(t))+2*C-1/4

DEVICE=PS,PYTH2.PS

C=1(1)50

pi=3.141592653589793

T=[YELLOW][line_width(0.48)],0.00001,pi/2,pi/1000

FRAME=0

SCALE=0.5,200.5 XLABEL= YLABEL= HEADER=

SIZE=1270,1270

XDIV=100,1200,100 YDIV=100,1200,100

Making a combined graph PYTH12.PS:

EPS JOIN PYTH12.PS PYTH1B.PS PYTH2.PS

___________________________________________________________________

For finding N(C)-values (before conceiving the general rule by consulting Sloane’s
Encyclopedia) I made a Survo macro (sucro) /NPYTH. For C = 864 it gives the fol-
lowing list of Pythagorean triples (X, Y, Z) for X < Y :

____________________________________________________________________

/NPYTH 864

Pythagorean triples for C=864: N(C)/2=42 triples found!

(2880,3024,4176) (2752,3186,4210) (2700,3264,4236) (2592,3456,4320)

(2496,3672,4440) (2457,3776,4505) (2376,4032,4680) (2304,4320,4896)

(2240,4644,5156) (2214,4800,5286) (2160,5184,5616) (2112,5616,6000)

(2052,6336,6660) (2016,6912,7200) (1984,7560,7816) (1971,7872,8115)

(1944,8640,8856) (1920,9504,9696) (1890,10944,11106) (1872,12096,12240)

(1856,13392,13520) (1836,15552,15660) (1824,17280,17376)

(1809,20160,20241) (1800,22464,22536) (1792,25056,25120)

(1782,29376,29430) (1776,32832,32880) (1764,43200,43236)

(1760,48384,48416) (1755,57024,57051) (1752,63936,63960)

(1746,84672,84690) (1744,95040,95056) (1740,126144,126156)

(1737,167616,167625) (1736,188352,188360) (1734,250560,250566)

(1732,374976,374980) (1731,499392,499395) (1730,748224,748226)

(1729,1494720,1494721)

Checking the result by editorial computing of Survo:

2*C^2=1492992

1492992(10:factors)=2^11*3^6 (11+1)*(6+1)/2=42

____________________________________________________________________

Here is the sucro code listing:

____________________________________________________________________

*TUTSAVE NPYTH

/ /NPYTH C

/ lists all Pythagorean triples (X<Y,Z) located on the hyperbola

/ Y = 2C(X-C)/(X-2C)

/ for any integer C.

/

/ def Wc=W1 Wx=W2 Wy=W3 Wz=W4 Wn=W5 Wt=W6

/

*{tempo -1}{R}{erase}{ref set 1}{R}{erase}{ref set 2}
/ Starting from maximum value of X+1:

*int((2+sqrt(2))*{print Wc})+1={act}{l} {save word Wx}{line start}{erase}
*{Wn=0}
/

+ A: {ref jump 1}{line start}{erase}{Wx=Wx-1}
- if Wx < 1 then goto E

*{print Wx}-2*{print Wc}={act}{l} {save word Wt}
- if Wt <= 0 then goto E

*2*{print Wc}*({print Wx}-{print Wc})/({print Wx}-2*{print Wc})={act}
*{l} {find .}{save char Wt}
/ If Wy is not integer, try a new X:

- if Wt ’=’ . then goto A

*{save word Wy}{line start}{erase}

*sqrt({print Wx}ˆ2+{print Wy}ˆ2)={act}{l} {find .}{save char Wt}
/ If Wz is not integer, try a new X:

- if Wt ’=’ . then goto A

/ A new triple is found. Add it to the list:

*{save word Wz}
* {line start}{erase}({print Wx},{print Wy},{print Wz}){line start}
*{save line Wt}
*{ref jump 2}{Wn=Wn+1}
*{write Wt}
* {ref set 2}
*{ref jump 1}{goto A}
/ Final remarks:

+ E: {ref jump 1}{erase}Pythagorean triples for C={print Wc}:
* N(C)/2={print Wn} triples found!

*{end}

____________________________________________________________________

I have also made a more extensive sucro P TRIPLE which plots the “tiny red flow-
ers” on the screen. The user can pick a “flower” by the mouse and the sucro tells
whether it corresponds to a Pythagorean triple. In any case it displays the X, Y,
and Z values. The procedure may be repeated. P TRIPLE is available in SURVO
MM (ver. 2.28+).

The current version of this paper can be downloded from
http://www.survo.fi/papers/pythagorean3.pdf

